Gradients of the Cartesian Components of the Oblate Magnetic Field of Arbitrary Direction

Markus Kraiger

mkraiger@student.tugraz.at

Bernhard Schnizer

schnizer@itp.tu-graz.ac.at

2011-04-29

Summary: In the general case the exterior potential of an oblate spheroid in an external homogeneous field depends on the constant vector, **n**, giving the symmetry axis, on the constant vector giving the field, on the position vector r = (x, y, z) in Cartesian coordinates and on three functions g1, g2, g3 depending on these coordinates through the functions

$$\text{wo} = \sqrt{\frac{4\,(\mathbf{n}.\mathbf{r})^2}{e^2} + \left(-1 + \frac{r^2}{e^2}\right)^2}$$

$$\text{up} = \sqrt{-1 + \frac{r^2}{e^2} + \sqrt{\frac{4\,(\mathbf{n}.\mathbf{r})^2}{e^2} + \left(-1 + \frac{r^2}{e^2}\right)^2}} = \sqrt{-1 + \frac{r^2}{e^2} + \text{wp}}\ .$$

When calculating the exterior magnetic field, gradients of g1, g2, g3 are needed. These are computed and simplified in this notebook. The simplifications are found by inspection of the initial form of the derivatives. The functions gi depend on two roots. Their derivatives contain again these roots multiplied by derivatives of the randicands which are quadratic polynomials. So one expects that the gradients will consist of such expressions. This aspect guides the guesses. These are finally checked by symbolic computation.
This notebook contains only the checks. The trials are in the unpublished notebook Oblatefq&dfq.nb .

e = eo = excentricity of the oblate spheroid.

M. Kraiger, B. Schnizer: Reaction Fields of Homogeneous Magnetic Spheroids of Arbitrary Direction in a Homogenous Magnetic Field. A Toolbox for MRI and MRS of Heterogeneous Tissue.
Report ITPR-2011-021, Institut fuer Theoretische Physik - Computational Physics, TU Graz, Austria.

$Version

"7.0 for Mac OS X x86 (64-bit) (February 19, 2009)"

Results of the Calculations

Input functions

$$wo = \sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2} + \left(-1 + \frac{x^2+y^2+z^2}{e^2}\right)^2}$$

$$uo = \sqrt{-1 + \frac{x^2+y^2+z^2}{e^2} + \sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2} + \left(-1 + \frac{x^2+y^2+z^2}{e^2}\right)^2}} = \sqrt{-1 + \frac{x^2+y^2+z^2}{e^2} + wo} = \sinh\eta$$

$$g1 = \mathrm{ArcCot}\left[\frac{\sqrt{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}}}{\sqrt{2}}\right] = \mathrm{arccot}\left[uo/\sqrt{2}\right]$$

$$= \mathrm{arccot}[\sinh\eta]$$

$$g2 = \frac{\sqrt{2}\sqrt{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}}}{1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}} = \frac{\sqrt{2}\;uo}{2 + uo\verb|^|2}$$

$$= \frac{\sinh\eta}{1+\sinh^2\eta} = \frac{\sinh\eta}{\cosh^2\eta}$$

$$g3 = \frac{\sqrt{2}}{\sqrt{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}}} = \frac{\sqrt{2}}{uo} = \frac{1}{\sinh\eta}$$

Output: The gradients of g1, g2, g3

ro

All gradients are proportional to the vector **rp**:

$$\mathbf{ro} = \left(\mathbf{r} + 2\,\frac{\mathbf{n}\,(\mathbf{n.r})}{uo^2}\right)$$

The gradients

$$\nabla f1 = \mathbf{dggq1} = -\frac{\sqrt{2}}{e^2}\,\mathbf{ro}\,\frac{uo}{wo\,(uo^2+2)}$$

$$\nabla f2 = \mathbf{dggq2} = \frac{\sqrt{2}}{e^2}\,\mathbf{ro}\,\frac{uo}{wo}\,\frac{2-uo^2}{(uo^2+2)^2}$$

$$\nabla f3 = \mathbf{dggq3} = -\frac{\sqrt{2}}{e^2}\,\mathbf{ro}\,\frac{1}{uo\;wo}$$

- **Only differences of the gradients are needed**

$$\nabla g1 - \nabla g2 = -\frac{\sqrt{2}}{e^2}\,ro\,\frac{4\,uo}{\left(2+uo^2\right)^2\,wo}$$

$$\nabla g1 - \nabla g3 = \frac{\sqrt{2}}{e^2}\,ro\,\frac{2}{uo\,\left(2+uo^2\right)\,wo}$$

Preliminary definitions

```
<< VectorAnalysis`
```

```
SetCoordinates[Cartesian[x, y, z]]
```

```
Cartesian[x, y, z]
```

```
vr = {x, y, z}
```

```
{x, y, z}
```

```
vn = {nx, ny, nz}
```

```
{nx, ny, nz}
```

```
vnr = vn.vr
```

$nx\,x + ny\,y + nz\,z$

```
r2 = vr.vr
```

$x^2 + y^2 + z^2$

```
e2 = e^2
```

e^2

```
vh0 = {H0x, H0y, H0z}
```

```
{H0x, H0y, H0z}
```

```
wo = Sqrt[(-1 + r2 / e2)^2 + 4 (vr.vn)^2 / e2]
```

$$\sqrt{\frac{4\,(nx\,x + ny\,y + nz\,z)^2}{e^2} + \left(-1 + \frac{x^2 + y^2 + z^2}{e^2}\right)^2}$$

```
uo = Sqrt[-1 + r2 / e2 + wo]
```

$$\sqrt{-1 + \frac{x^2 + y^2 + z^2}{e^2} + \sqrt{\frac{4\,(nx\,x + ny\,y + nz\,z)^2}{e^2} + \left(-1 + \frac{x^2 + y^2 + z^2}{e^2}\right)^2}}$$

```
vnn = {1/√2, 1/√3, 1/√6}
```

$$\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{6}}\right\}$$

```
ro = (vr + 2 vn (vn.vr)/uo^2)
```

$$\left\{x+\frac{2\,nx\,(nx\,x+ny\,y+nz\,z)}{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}},\right.$$
$$y+\frac{2\,ny\,(nx\,x+ny\,y+nz\,z)}{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}},$$
$$\left.z+\frac{2\,nz\,(nx\,x+ny\,y+nz\,z)}{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}}\right\}$$

- **Numeric values for tests**

```
vnn = {1/√2, 1/√3, 1/√6};
vnn.vnn
```

1

```
tev = {x → 1.1, y → 1.7, z → 2.3, e → 1.73, Thread[ vn → vnn]} // Flatten
```

$$\left\{x\to 1.1,\ y\to 1.7,\ z\to 2.3,\ e\to 1.73,\ nx\to \frac{1}{\sqrt{2}},\ ny\to \frac{1}{\sqrt{3}},\ nz\to \frac{1}{\sqrt{6}}\right\}$$

fq1, ∇f1

```
gq1 = ArcCot[uo/√2]
```

$$\text{ArcCot}\left[\frac{\sqrt{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}}}{\sqrt{2}}\right]$$

- **dgq1 = ∇g1**

```
dgq1 = Grad[gq1];
```

After some manipulations one finds:

$$\text{dggq1} = -\frac{\sqrt{2}}{e^2}\ \text{ro}\ \frac{\text{uo}}{\text{wo}\,(\text{uo}^2+2)};$$

```
dgq1 - dggq1;
Simplify[%]
```

{0, 0, 0}

fq2, ∇f2

```
gq2 = Sqrt[2] uo / (uo^2 + 2)
```

$$\frac{\sqrt{2}\sqrt{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}}}{1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}}$$

- **dgq2 = ∇g2**

```
dgq2 = Grad[gq2];
```

After some manipulations one finds:

$$\text{dggq2} = \frac{\sqrt{2}}{e^2}\,\text{ro}\,\frac{\text{uo}}{\text{wo}}\,\frac{2-\text{uo}^2}{\left(\text{uo}^2+2\right)^2};$$

```
dgq2 - dggq2;
Simplify[%]
{0, 0, 0}
```

fq3, ∇fq3

```
gq3 = Sqrt[2] / uo
```

$$\frac{\sqrt{2}}{\sqrt{-1+\frac{x^2+y^2+z^2}{e^2}+\sqrt{\frac{4\,(nx\,x+ny\,y+nz\,z)^2}{e^2}+\left(-1+\frac{x^2+y^2+z^2}{e^2}\right)^2}}}$$

- **dgq3 = ∇g3**

```
dgq3 = Grad[gq3];
```

After some manipulations one finds:

$$\text{dggq3} = -\frac{\sqrt{2}}{e^2}\,\text{ro}\,\frac{1}{\text{uo wo}};$$

```
dgq3 - dggq3;
Simplify[%]
{0, 0, 0}
```

∇gq1 - ∇gq2, ∇gq1 - ∇gq3

The differences of the gradients are best computed by working with the ro, uo, wo.
For that reason all the symbols defined up to now are cleared.

```
Clear["Global`*"]
```

$$\text{dggq1} = -\frac{\sqrt{2}}{e^2}\,\text{ro}\,\frac{\text{uo}}{\text{wo}\left(\text{uo}^2+2\right)}$$

$$-\frac{\sqrt{2}\ \text{ro uo}}{e^2\left(2+\text{uo}^2\right)\text{wo}}$$

$$\text{dggq2} = \frac{\sqrt{2}}{e^2}\,\text{ro}\,\frac{\text{uo}}{\text{wo}}\,\frac{2-\text{uo}^2}{\left(\text{uo}^2+2\right)^2}$$

$$\frac{\sqrt{2}\ \text{ro uo}\left(2-\text{uo}^2\right)}{e^2\left(2+\text{uo}^2\right)^2\text{wo}}$$

$$\text{dggq3} = -\frac{\sqrt{2}}{e^2}\,\text{ro}\,\frac{1}{\text{uo wo}}$$

$$-\frac{\sqrt{2}\ \text{ro}}{e^2\ \text{uo wo}}$$

```
dggq1 - dggq2 // FullSimplify
```

$$-\frac{4\sqrt{2}\ \text{ro uo}}{e^2\left(2+\text{uo}^2\right)^2\text{wo}}$$

```
dggq1 - dggq3 // FullSimplify // Factor
```

$$\frac{2\sqrt{2}\ \text{ro}}{e^2\ \text{uo}\left(2+\text{uo}^2\right)\text{wo}}$$